
4.3 Script com o conjunto de comandos de integração . . . . . . . 26
4.4 Detalhamento da função integra ate choque . . . . . . . . . . 29
4.5 Para executar o script . . . . . . . . . . . . . . . . . . . . . . 29

5 Exemplo: choque com integração aproximada 31
5.1 As equações do movimento . . . . . . . . . . . . . . . . . . . . 32
5.2 Script para simulação do movimento . . . . . . . . . . . . . . 33

Caṕıtulo 1

Introdução

Nesta apostila descreve-se, através de alguns exemplos, como integrar no
tempo equações de movimento de sistemas de corpos ŕıgidos. Recomenda-se
a elaboração de dois arquivos, scripts, diferentes. Um deles funciona como um
programa principal que invoca o integrador ( a função ode(.);), que por sua
vez, invoca uma função que contém as equações do movimento. O programa
principal será um arquivo ASCII com um nome arbitrário mas que deve
ter terminação .sci. A função com as equações do movimento será descrita
em um arquivo ASCII com nome arbitrário mas que deve ter terminação
.sce. A recomendação do uso de dois arquivos é questão de estilo e pode ser
abandonada.

Na disciplina PME2200 Mecânica B o aluno desenvolve habilidades para
determinar as equações do movimentos de sistemas de corpos ŕıgidos, seja
através dos teoremas do movimento do centro de massa e da quantidade
de movimento angular ou através das equações de Lagrange. A integração
numérica das equações do movimento permite visualizar a resposta destes
sistemas, trazendo significado f́ısico a cada termo de cada equação. É, por-
tanto, poderoso instrumento didático para transmitir o poder informativo das
equações do movimento.

1.1 O integrador

O integrador no Scilab para problemas de valores iniciais chama-se ODE. A
maneira mais simples de invocar o integrador é

y=ode(t0,y0,t,g);

onde t0 é o instante inicial de integração, y0 é o vetor de estado inicial, t é o
vetor de instantes em que se deseja a resposta do sistema e g é a função a ser

6 CAPÍTULO 1. INTRODUÇÃO

integrada. Noutras palavras, t0 e y0 são as condições iniciais de um sistema
equações diferenciais do tipo

dy

dt
= g(t, y) (1.1)

onde a função g pode ser uma função escalar ou um vetor de funções conforme
a eq. 1.2,

g(t, y) =











g1(t, y1, y2, . . . , yn)
g2(t, y1, y2, . . . , yn)

. . .
gn(t, y1, y2, . . . , yn)











, (1.2)

quando g(t, y) é um vetor de funções, y é um vetor de mesma dimensão. Para
poder invocar este integrador é preciso definir as condições iniciais, o vetor t
e a função que será integrada.

1.2 Leituras recomendadas

Informações adicionais podem ser encontradas nos seguintes śıtios

www.dca.ufrn.br/ pmotta/sciport-3.0.pdf
www.dca.ufrn.br/ pmotta/scilab-slides3.pdf
http://cermics.enpc.fr/scilab new/site/Liens/intro/intro.html

1.3 Autoria dos exemplos

O exemplo do véıculo passando por uma lombada foi desenvolvido por di-
versas pessoas, entre as pessoas que formularam o problema encontram-se o
Prof. Dr. Roberto Martins de Souza, a Dra. Sara Aida Rodriguez Pulecia e
o autor deste tutorial. O exemplo da part́ıcula que colide com solo foi inspi-
rado da apostila tutorial do ambiente gráfico de simulação e com a ajuda do
doutorando Marcos Rabelo. O exemplo do anel impulsionado por uma barra
que está girando é de autoria dos professores Celso Pupo Pesce e Flávio Celso
Trigo. A equipe de professores de Mecânica B agradece a possibilidade de
reprodução destes exemplos.



Caṕıtulo 2

Exemplo: duas massas, duas
molas e dois amortecedores

Problemas de valores iniciais definidos por conjuntos de equações diferenci-
ais ordinárias podem ser integrados utilizando o Scilab e linhas de comandos.
Neste tutorial desenvolve-se um exemplo de integração no tempo de um sis-
tema com duas massas, duas molas e dois amortecedores, conforme a fig. 2.1.

c c

kk

m m

x2 x1

sen(t)

Figura 2.1: Sistema com duas massas, duas molas e dois amortecedores

Decorre da Segunda Lei de Newton e do modelo de amortecimento viscoso,

mẍ1 = −k(x1 − x2)− c(ẋ1 − ẋ2) + sin(t) (2.1)

e

mẍ2 = −kx2 − cx2 + k(x1 − x2) + c(ẋ1 − ẋ2) (2.2)

Rearrajando estas duas equações para isolar as acelerações do lado es-
querdo, temos,

ẍ1 = −
k

m
(x1 − x2)−

c

m
(ẋ1 − ẋ2) + sin(t) (2.3)

e

ẍ2 =
k

m
(x1 − 2x2) +

c

m
(ẋ1 − 2ẋ2). (2.4)

Este sistema de duas equações diferenciais de segunda ordem deve ser
transformado em um sistema de quatro equações diferenciais de primeira
ordem para que possam ser utilizados integradores clássicos. Para promover
esta transformação define-se um vetor y em função dos deslocamentos e das
derivadas,

y =



















y1
y2
y3
y4



















=



















x1

ẋ1

x2

ẋ2



















(2.5)

e sua derivada no tempo ẏ torna-se

ẏ =



















ẏ1
ẏ2
ẏ3
ẏ4



















=



















ẋ1

ẍ1

ẋ2

ẍ2



















(2.6)

Em termos deste vetor de estado, a eq. 2.3 e a eq. 2.4 podem ser repre-
sentadas por



















ẋ1

ẍ1

ẋ2

ẍ2



















=



















ẋ1

−
k
m
(x1 − x2)−

c
m
(ẋ1 − ẋ2) + sin(t)
ẋ2

k
m
(x1 − 2x2) +

c
m
(ẋ1 − 2ẋ2)



















(2.7)

2.1 Definindo o conjunto de equações dife-

renciais no Scilab

Uma das maneiras de definir um conjunto de equações diferenciais no Scilab
é através de um arquivo texto, preparado em um editor de textos comum,
e que recebe o sufixo .sci, por convenção. Assim, o conjunto de equações
do nosso exemplo, eq. 2.7, foi definido pelo arquivo exemplo.sci, reproduzido
nas linhas que seguem

function [dy]=g(t,y)

y1=y(2);

y2=-k*(y(1)-y(3))/m-c*(y(2)-y(4))/m+sin(t);

y3=y(4);

y4=k*(y(1)-2*y(3))/m+c*(y(2)-2*y(4))/m;

dy=[y1;y2;y3;y4];

endfunction

Resta ainda escrever um script onde se definem as constantes m, k, c, o
vetor de instantes em que se deseja a resposta e como deve ser o gráfico que
mostra a resposta.

2.2 Script com o conjunto de comandos de

integração

Nesta seção descreve-se um arquivo com os comandos necessários à inte-
gração da função g(t, y). Este arquivo de comandos, exemplo.sce, tem, por
convenção, sufixo .sce. Inicialmente definem-se as variáveis m, k e c. Segue
uma leitura da função g(t, y). Definem-se as condições iniciais t inicial e y0.
Definem-se ainda o incremento de tempo de integração Dt, o instante final
de integração t final e o vetor de instantes em que se deseja a resposta do
sistema y(t).

// Rigidez em N/m

k=1.0;

// Coeficiente de amortecimento em Ns/m

c=0.001;

// Massa em kg

m=1.0;

// Carrega a definicao da funcao dy=g(t,y)

exec exemplo.sci;

// Vetor estado inicial

y0=[0;0;0;0];

// Instante inicial de integracao

t_inicial=0.0;

// Incremento de tempo desejado na resposta

Dt=0.01;

// Instante final de integracao

t_final=16.0;

// Vetor de instantes desejados na resposta

t=[t_inicial:Dt:t_final];

// Rotina de integracao da funcao dy=g(t,y)

y=ode(y0,t_inicial,t,g);

// Comando para gerar grafico dos deslocamentos

scf(0)

plot2d(t’,[y(1,:)’,y(3,:)’],leg="x1@x2");

xtitle(’Deslocamento das massas’,’tempo (s)’,’Deslocamento (m)’);

// Guarda o grafico em arquivo eps

filename=’deslocamento’;

xs2eps(0,filename);

Os últimos comandos promovem a integração numérica propriamente
dita, geram um gráfico dos deslocamentos das duas massas , fig. 2.2, e criam
um arquivo em Encapsulated Postscript, deslocamento.eps.1

1No comando plot2d o sinal de apóstrofe indica transposta de vetor ou matriz, ou seja,
t
′

é a transposta de t. Outro detalhe que merece explicações é o termo y(1, :), leia-se tome

todas as colunas da primeira linha da matriz y.

Figura 2.2: Deslocamento calculado das duas massas

2.3 Para executar o script

Após invocar o Scilab através de seu ı́cone, aparece uma tela com um prompt

que é o śımbolo −− >. A primeira providência a ser tomada é descobrir qual
é o diretório de trabalho atual. Para conhecer o diretório de trabalho atual
deve-se digitar no prompt pwd e pressionar a tecla <enter >. Se acaso os
arquivos de comandos e de definição da função, exemplo.sce e exemplo.sci,
não se encontram no diretório de trabalho, deve-se alterar o diretório de
trabalho através do comando chdir(). Utilize o help do scilab para enten-
der o comando chdir() no particular sistema operacional que você estiver
utilizando.

Finalmente, para executar o script de comandos que se encontra no di-
retório de trabalho, exemplo.sce, basta digitar

exec(‘‘exemplo.sce’’);

ou simplesmente

exec exemplo.sce;



Caṕıtulo 3

Exemplo: véıculo 2D passando
por lombada

O sistema mostrado na fig. 3.1 representa uma simplificação de um véıculo
e seu sistema de suspensão. Nesta simplificação, o sistema é composto por
um sólido retangular de massa M e por duas massas concentradas m1 e m2.
O ponto P indica o local onde o condutor do véıculo está posicionado. O
sólido retangular está apoiado sobre dois conjuntos mola-amortecedor, cada
um dos quais com valores próprios de rigidez da mola k e da constante c
do amortecedor viscoso linear. As duas molas têm comprimento l0 quando
a deformação é nula. No instante mostrado na fig. 3.1, o conjunto move-se
sobre uma superf́ıcie horizontal, com velocidade constante ~v = v~i.

Figura 3.1: Modelo 2D de véıculo e via com lombada

Após percorrer uma distância L1, o véıculo tem de suplantar um obstáculo
em sua trajetória, como indicado na fig. 3.2. Durante a passagem pelo
obstáculo, o véıculo permanece com velocidade constante . O pavimento
tem altura definida por:

h(x) =















0 se x≤L1 + L/2
H
2

(

1− cos 2π
L2

x
)

se L1 + L/2 < x < L1 + L2 + L/2

0 se x≥L1 + L2 + L/2















(3.1)

Figura 3.2: Véıculo sobre a lombada

Admitem-se como dados X0 = 0; L = 5 m; L1 = 50 m; L2 = 3, 7 m;
H = 0, 1 m; b = 0, 1 m; b1 = 0, 7 m; M = 400 kg; V = 18 m/s e g = 9, 8
m/s2. Sugere-se utilizar 0, 001 s como intervalo de integração. Considera-
se ainda m1 = m2 = 120 kg; d1 = d2 = 2, 5 m; k1 = k2 = 30.000 N/m;
c1 = c2 = 2.100 Ns/m ; yO(0) = l0 = 0, 3 m e α(0) = 0.

As equações do movimento foram determinadas para o movimento de
arfagem α, ver eq. 3.2. A dedução pode ser feita através dos teoremas do
movimento do centro de massa e do teorema da quantidade de movimento
angular ou através da Mecânica Anaĺıtica.

sen(α)(M2d2 −M1d1)ẍO

+cos(α)(M1d1 −M2d2)ÿO
+(M1d

2

1
+M2d

2

2
)α̈

+cos(α)(M1d1 −M2d2)g
−k1L cos (α)(l0 − yO − Lsen(α) + yA(t))

+k2L cos (α)(l0 − yO + Lsen(α) + yB(t)) =
−c1L cos (α)(ẏO + Lα̇ cos (α))− ẏA(t))
+c2L cos (α)(ẏO − Lα̇ cos (α))− ˙yB(t))

(3.2)

e para a oscilação vertical yO, ver eq. 3.3.

(M1 +M2 +M +m)ÿO
+cos(α)(M1d1 −M2d2)α̈
+sen(α)(M2d2 −M1d1)α̇

2

+(M1 +M2 +M +m)g
−k1(l0 − yO − Lsen(α) + yA(t))

−k2(l0 − yO + Lsen(α) + yB(t)) =
−c1(ẏO + Lα̇ cos (α))− ẏA(t))
−c2(ẏO − Lα̇ cos (α))− ẏB(t))

(3.3)

3.1 Definindo o conjunto de equações dife-

renciais no Scilab

Uma das maneiras de definir um conjunto de equações diferenciais no Scilab é
através de um arquivo texto, preparado em um editor de textos comum, e que
recebe o sufixo .sci, por convenção. Assim, o conjunto de equações do nosso
exemplo, eq. 3.3 e eq. 3.2, foi definido pelo arquivo emsc.sci, reproduzido no
script que segue. Observa-se que a equação, eq. 3.1, que descreve a altura
da via h(x) foi codificada no ińıcio deste arquivo, junto com as equações
diferenciais, ver linhas 2 a 25 do arquivo emsc.sci. A última linha define o
vetor ẋ, que no script recebe o nome de dy.

function [dy]=car(t,y)

if t<tA1 then

yA=0.0;

dyA=0.0;

else

if t<tA2 then

yA=H*(1-cos(2*pi*(v*t-L1)/L2))/2;

dyA=H*(sin(2*pi*(v*t-L1)/L2))*pi*v/L2;

else

yA=0.0;

dyA=0.0;

end

end

if t<tB1 then

yB=0.0;

dyB=0.0;

else

if t<tB2 then

yB=H*(1-cos(2*pi*(v*t-L1-L)/L2))/2;

dyB=H*(sin(2*pi*(v*t-L1-L)/L2))*pi*v/L2;

else

yB=0.0;

dyB=0.0;

end

end

F=-A*g+B*sin(y(3))*y(4)*y(4)+k1*(L0-y(1)

-L*sin(y(3))/2+yA)+k2*(L0-y(1)+L*sin(y(3))/2+yB)

-c1*(y(2)+L*y(4)*cos(y(3))/2-dyA)-c2*(y(2)-

L*y(4)*cos(y(3))/2-dyB);

H=-B*g*cos(y(3))+k1*L*cos(y(3))*(L0-y(1)-L*sin(y(3))/2+yA)/2

-k2*L*cos(y(3))*(L0-y(1)+L*sin(y(3))/2+yB)/2

-c1*L*cos(y(3))*(y(2)+L*y(4)*cos(y(3))/2-dyA)/2

+c2*L*cos(y(3))*(y(2)-L*y(4)*cos(y(3))/2-dyB)/2;

determinante=A*C-B*B*cos(y(3))*cos(y(3));

dy1=y(2);

dy2=(C*F-B*cos(y(3))*H)/determinante;

dy3=y(4);

dy4=(A*H-B*cos(y(3))*F)/determinante;

dy5=dy2;

dy=[dy1;dy2;dy3;dy4;dy5];

endfunction

3.2 Script com o conjunto de comandos de

integração

Nesta seção descreve-se um arquivo com os comandos necessários à inte-
gração da função g(t, y). Este arquivo de comandos, exemplo.sce, tem, por

convenção, sufixo .sce. Inicialmente definem-se as variáveis, por exemplo, m,
m1, m2, k1 e c1. Segue uma leitura da função g(t, y). Definem-se as condições
iniciais tinicial e y0. Definem-se ainda o incremento de tempo de integração
Dt, o instante final de integração tfinal e o vetor de instantes em que se deseja
a resposta do sistema y(t).

clear all

k1=30000;

k2=30000;

d1=2.50;

d2=2.50;

M=400;

M1=120;

M2=120;

L=5.0;

L1=90.0;

L2=3.7;

L0=0.3;

Y_0=L0;

b=0.1;

c1=2100.0;

c2=2100.0;

v=48.0;

pi=3.1416;

g=9.8;

H=0.1;

b1=0.7;

y0=[L0; 0.0; 0.0; 0.0; 0.0];

JO=M*(b*b+L*L)/12;

A=M1+M2+M;

B=(M1*d1-M2*d2);

C=M1*d1*d1+M2*d2*d2+M*(b*b+L*L)/12;

tA1=L1/v

tA2=(L1+L2)/v

tB1=(L1+L)/v

tB2=(L1+L2+L)/v

t_inicial=0.0;

Dt=0.001;

t_final=10.0;

t=[t_inicial:Dt:t_final];

// inicializa funcao

exec emsc3.sci;

ys=ode(’stiff’,y0,t_inicial,t,car);

scf(1)

plot2d(t’,[ys(1,:)’]);

xtitle(’Deslocamento vertical do ponto O vs. tempo’,’t’,’Yo’);

scf(2)

plot2d(t’,[ys(3,:)’]);

xtitle(’Arfagem vs. tempo’,’t’,’Angulo alfa’);

for i=tA1/Dt:t_final/Dt

yb1(i)=ys(1,i)+b1*sin(ys(3,i));

end

for i=1:tA1/Dt

yb1(i)=ys(1,tA1/Dt)+b1*sin(ys(3,tA1/Dt));

end

desconforto=max(abs(yb1(:)-yb1(t_final/Dt-1)))

scf(3)

plot2d(yb1);

xtitle(’Altura na posiç~ao do piloto vs. tempo’,’t’,’Yb1 (m)’);

Três gráficos são gerados, na primeira tela scf(1) está a história temporal
do deslocamento vertical variável yo, na segunda tela scf(2) está a história
temporal do ângulo de arfagem α e na terceira tela scf(3) está o deslocamento
vertical no assento do piloto yb1, que depende tanto de α quanto de y0. A
fig. 3.3 apresenta o gráfico de yO = yO(t) e a fig. 3.4 apresenta o gráfico
de αO = αO(t) para esta condição. Observe que o gráfico de y0 permite
identificar dois instantes distintos do movimento, um deles, inicial, associado
à estabilização de yO em um valor de equiĺıbrio e outro associado à passagem

do véıculo pelo obstáculo.

Figura 3.3: Véıculo sobre a lombada

Figura 3.4: Véıculo sobre a lombada

Desta forma é posśıvel obter informações sobre conforto ou segurança de
um véıculo na fase de projeto, lembrando que as hipóteses de modelagem
restringem o campo de perguntas que podemos dirigir à simulação numérica,
a metodologia restringe o campo das conclusões pertinentes.

3.3 Para executar o script

Após invocar o Scilab através de seu ı́cone, aparece uma tela com um prompt

que é o śımbolo −− >. A primeira providência a ser tomada é descobrir
qual é o diretório de trabalho atual. Para conhecer o diretório de trabalho
atual deve-se digitar no prompt pwd e pressionar a tecla <enter >. Se acaso
os arquivos de comandos e de definição da função, emsc.sce e emsc.sci, não
se encontram no diretório de trabalho, deve-se alterar o diretório de traba-
lho através do comando chdir(). Utilize o help do scilab para entender o
comando chdir() no particular sistema operacional que você estiver utili-
zando.

Finalmente, para executar o script de comandos que se encontra no di-
retório de trabalho, emsc.sce, basta digitar

exec(‘‘emsc.sce’’);

ou simplesmente

exec emsc.sce;

Caṕıtulo 4

Exemplo: part́ıcula que colide
contra o solo

Considere uma part́ıcula de massa m que é abandonada de uma altura h0 do
solo com velocidade vertical v0, esta velocidade é considerada positiva quando
a part́ıcula se afasta do solo. Admite-se ainda que a aceleração da gravidade
é g = 9, 81 m/s2 e que o coeficiente de restituição da colisão part́ıcula/solo é
e = 0, 85. Nestas condições, a part́ıcula irá colidir diversas vezes contra solo
conforme a figura 4.1

23

Figura 4.1: Part́ıcula que se choca contra o solo

4.1 As equações do movimento

Do teorema do movimento do baricentro, podemos afirmar sobre a part́ıcula,

mÿ = −mg (4.1)

que pode ser reduzida a

ÿ = −g. (4.2)

Na maioria dos integradores numéricos o sistema de equações de movi-
mento deve ser informado na forma,

ẋ = f(x, t) (4.3)

portanto, como um sistema de equações diferenciais ordinárias de primeira
ordem. A eq. 4.2, que é de segunda ordem, pode ser transformada em um

sistema de equações de primeira ordem através de um artif́ıcio. Define-se um
vetor estado, por exemplo,

x =

{

y
ẏ

}

. (4.4)

E a primeira derivada temporal do vetor estado é

ẋ =

{

ẏ
ÿ

}

. (4.5)

A equação 4.2 pode ser escrita em termos do vetor estado e sua derivada
como

{

y
ẏ

}

=

[

0 1
0 0

]{

y
ẏ

}

+

[

0
−1

]

g (4.6)

ou em forma matricial mais compacta

ẋ = Ax+Bg (4.7)

onde a matriz A é

A =

[

01
00

]

(4.8)

e a matriz B é

B =

[

0
−1

]

(4.9)

A função de Scilab com as equações do movimento é representada no
arquivo choque teste.sci reproduzido abaixo.

function dy=f(t,y)

dy = A*y+B;

endfunction

4.2 O critério algébrico para encerrar a inte-

gração numérica

O integrador de linha de comando que o Scilab oferece chama-se ode().
Ele permite interromper a integração numérica caso uma ou mais funções
algébricas se tornem zero. Estas funções algébricas são informadas ao integra-
dor ode() através da definição de uma função de Scilab, aqui neste exemplo,

no arquivo choque condicao.sci, reproduzido a seguir. Este arquivo informa
que no instante em que a altura z1 = y for pequena e a velocidade z2 = ẏ for
negativa, a condição de colisão foi satisfeita e a integração deve ser interrom-
pida para que novas condições de velocidade sejam impostas como resultado
do choque.

function z=g(t,y)

z1=[1 0]*y;

z2=[0 1]*y;

z3=z1+sqrt(z2*z2)+z2;

z=[z3];

endfunction

4.3 Script com o conjunto de comandos de

integração

Sempre que inicializado o integrador ode() responde com uma lista de in-
formações, uma matriz com a história de estados Y nos instantes de tempo
definidos previamente pelo vetor temp e com uma matriz linha rd, cujo pri-
meiro elemento contém o instante de tempo em que a condição de colisão
foi satisfeita e o segundo elemento contém o número de condições algébricas
que resultaram satisfeitas e, portanto, têm valor zero. No nosso exemplo há
apenas uma condição algébrica definida, a função g é escalar.

Na primeira parte do script os parâmetros do problema são informadas,
como variáveis globais. Define-se o tempo inicial, o tempo final de integração,
o intervalo de tempo em que gostaŕıamos de conhecer o estado do sistema Dt,
a altura inicial da part́ıcula h0, a velocidade inicial da part́ıcula v0, o estado
inicial y0, o coeficiente de restituição e a aceleração da gravidade grav.

// Exemplo de Choque de uma partı́cula com o solo

clc

global temp estado

//Instante inicial de integracao [s])

t_inicial = 0.0;

//Incremento de tempo na resposta [s]

Dt = 0.01;

//Instante final de integraç~ao [s]

t_final = 7.0;

//Altura inicial [m]

h0 = 5;

//Velocidade inicial [m/s]

v0 = -1;

//Estado Inicial

y0=[h0; v0];

//Coeficiente de restituiç~ao

e = 0.85;

//Aceleraç~ao da gravidade [m/s^2]

grav = 9.81;

A seguir são introduzidas as matrizes A e B que pertencem às equações
do movimento, eq. 4.7 e através do comando getd, todas as funções do di-
retório especificado são lidas e incorporadas para serem utilizadas na sessão
de Scilab. As funções definidas no diretório são f , g e integra ate choque
(esta última função será abordada em breve).

//Carrega a definiç~ao da funç~ao dy = f(t,y)

A=[0 1; 0 0];

B=[ 0; -grav];

getd("/home/raul/disciplinas/pme2200/choque");

Como o integrador deve ser interrompido na condição de colisão, g(x, t) =
0, é necessário ir armazenando os estados já calculados em uma matriz perene,
que conterá a história dos estados de diversos peŕıodos sem colisão. Definem-
se então dois vetores, temp e a matriz estado, onde cada coluna está associada
a um instante múltiplo do intervalo de resposta Dt. A matriz é inicializada
com zeros, tanto na primeira linha que representa a história do deslocamento
y, quanto na segunda linha que representa a história da velocidade ẏ.

//inicia o tempo

t0=t_inicial;

// aloca memoria para vetor tempo e matriz historia do estado

n_tempo=floor(t_final/Dt);

for i=1:n_tempo

temp(i)=(i-1)*Dt;

vetor_zero(i)=0.0;

end

estado=[vetor_zero’; vetor_zero’];

Procede-se, então, uma sequência de integrações numéricas que são inter-
rompidas pelo critério de colisão. Após cada colisão, as condições de veloci-
dade pós-choque são impostas pela função integra ate choque. A sequência
de integrações numéricas são gerenciadas pelo laço (loop) de programação
while que segue. Se a integração for interrompida por outro motivo, dife-
rente da condição de colisão, então a altura da part́ıcula é diferente de zero e
interrompe-se o laço de programação while. Neste laço, os estados integrados
numericamente são armazenados na matriz estado.

t_choque=0.0;

while t_choque<t_final do

[t_choque,y_choque,estado]=integra_ate_choque(t_choque,t_final,Dt,y0,estado);

y0=y_choque

if abs(y_choque(1,1))>1.0e-8 then

t_choque=t_final

end

end

Quando a sequência de integrações numéricas estiver terminada, elabora-
se um gráfico da história da altura e arquiva-se este gráfico no formato jpg,
neste exemplo.

scf(0);

xtitle(’Altura (m) vs. Tempo (s)’);

plot2d(temp’,estado(1,:));

//Exportando para jpg

filename=’choque_resultado.jpg’;

xs2jpg(0,filename);

Resta detalhar como opera a função integra ate choque, que é uma função
definida pelo usuário.



4.4. DETALHAMENTO DA FUNÇÃO INTEGRA ATE CHOQUE 29

4.4 Detalhamento da função integra ate choque

Esta função invoca o integrador ode do Scilab, de um certo instante t inic, até
que ocorra uma colisão ou até que seja atingido o tempo final t final. Deseja-
se que os estados resultantes da integração numérica sejam reportados em
instantes múltiplos de Dt. A função ode devolve uma matriz Y e uma matriz
linha rd que contém no primeiro elemento o instante de colisão t choque. O
último estado (última coluna) da matriz Y contém o estado imediatamente
antes do choque. O estado imediatamente antes do choque, Y (:, coluna),
tem sua velocidade modificada através de [10; 0 − e] ∗ Y (:, coluna). O ve-
tor dscrtime é um sub-vetor do vetor temp e contém instantes de tempo
múltiplos de Dt.

function [t_choque,y_choque,estado]=integra_ate_choque(t_inic,t_final,Dt,y0,estado)

n=2; //numero de componentes do vetor estado

t_inic_dt=t_inic-modulo(t_inic,Dt)+Dt;

dscrtime = t_inic_dt:Dt:t_final;

[Y,rd]=ode("roots",y0,t_inic_dt,dscrtime,f,1,g);

t_choque=rd(1,1);

[linha,coluna]=size(Y);

y_choque=[1 0; 0 -e]*Y(:,coluna);

Finalmente, os estados obtidos nesta particular integração numérica são
armazenados no vetor global e perene estado para que, posteriormente, seja
posśıvel elaborar um gráfico contendo múltiplas colisões.

n_t_discreto=floor(t_choque/Dt);

n_t_discreto_old=floor(t_inic/Dt);

for i=n_t_discreto_old+1:n_t_discreto+1

for j=1:n

estado(j,i)=Y(j,i-n_t_discreto_old);

end

end

endfunction

4.5 Para executar o script

Após invocar o Scilab através de seu ı́cone, aparece uma tela com um prompt

que é o śımbolo −− >. A primeira providência a ser tomada é descobrir qual
é o diretório de trabalho atual. Para conhecer o diretório de trabalho atual
deve-se digitar no prompt pwd e pressionar a tecla <enter >. Se acaso os

arquivos de comandos e de definição da função, emsc.sce e emsc.sci, não se
encontram no diretório de trabalho, deve-se alterar o diretório de trabalho
através do comando chdir(). Utilize o help do Scilab para entender o co-
mando chdir() no particular sistema operacional que você estiver utilizando.
Finalmente, para executar o script de comandos que se encontra no diretório
de trabalho, emsc.sce, basta digitar

exec(‘‘teste_looping.sce’’);

ou simplesmente

exec teste_looping.sce;



Caṕıtulo 5

Exemplo: choque com
integração aproximada

Neste exemplo, formulado pelo Prof. Celso Pupo Pesce e pelo Prof. Flávio
Celso Trigo, recomenda-se implementar a simulação de dinâmica com even-
tuais choques através de uma estratégia que aproxima o instante de choque.
O erro sobre o instante de choque pode ser diminuido ao diminuir o inter-
valo de integração Dt. Uma maneira mais precisa de detectar o instante de
choque foi descrita no caṕıtulo 4.

Considere o sistema mostrado na figura 5.1, em que uma barra homogênea
e delgada AB de massa m e comprimento L é vinculada a um bloco de massa
M e comprimento 2a, através de uma articulação ideal em C. Este conjunto
pode deslizar sem atrito sobre a guia horizontal entre os batentes P e Q, fixos
nesta guia e separados por uma distância 4r. Uma mola ideal que possui
comprimento livre 2r e constante elástica K é acoplada ao bloco e à base
fixa W . O coeficiente de restituição entre o bloco de massa M e os batentes
P e Q vale e. A posição do bloco de massa M é dada por x(t). O parâmetro
β pode ser ajustado para modificar a proporção entre os segmentos BC e AC
da barra.

Efetue simulações, por exemplo, durante 12 segundos utilizando os parâmetros
M = 1, 0 kg, m = 9, 0 kg, L = 1, 5 m, r = 50 mm, a = 25 mm, g = 10 m/s2;
K = 100 N/m, e = 1, 0, β = 0, 25, x(0) = 2r, ẋ(0) = 0, 0, θ(0) = π/3 e
θ̇(0) = 0, 0.

31

Figura 5.1: Barra articulada em anel

5.1 As equações do movimento

Do teorema do movimento do baricentro e do teorema da quantidade de
movimento angular podem ser deduzidas as equações do movimento. São
duas equações diferenciais acopladas de segunda ordem.

Na maioria dos integradores numéricos o sistema de equações de movi-
mento deve ser informado na forma,

η̇ = f(η, t) (5.1)

portanto, como um sistema de equações diferenciais ordinárias de primeira
ordem. As equações de movimento que são de segunda ordem, podem ser

transformadas em um sistema de 4 equações de primeira ordem através de
um artif́ıcio algébrico. Define-se um vetor estado, por exemplo,

η =



















x
ẋ
θ

θ̇



















. (5.2)

E a primeira derivada temporal do vetor estado é

η̇ =























ẋ
ẍ

θ̇

θ̈























. (5.3)

A equação 4.2 pode ser escrita em termos do vetor estado e sua derivada,
em forma vetorial, como

η̇ = f(η, t) (5.4)

onde f é um vetor de funções que dependem de η e t.
A função de Scilab com as equações do movimento é representada no

arquivo choque teste.sci reproduzido abaixo.

function dx=f(t,x)

detM= Jc*(M+m)-m*m*d*d*cos(x(3))*cos(x(3));

// equacoes do movimento dx/dt= f(t,x)

// onde x eh o vetor estado definido por

// x = [ x; dx/dt; teta; d(teta)/dt];

dz1= x(2);

dz2= (Jc*(-m*x(4)*x(4)*d*sin(x(3))-k*(x(1)-r))+m*d*cos(x(3))*(-m*grav*d*sin(x(3))))

dz3= x(4);

dz4= (m*d*cos(x(3))*(-m*x(4)*x(4)*sin(x(3))*d-k*(x(1)-r))+(M+m)*(-m*grav*d*sin(x(3)

dx = [dz1; dz2; dz3; dz4];

endfunction

as constantes M, m, Jc, d, grav, k e r estão definidos como variáveis globais
no arquivo simula choque.sce, reproduzido abaixo.

5.2 Script para simulação do movimento

// Codigo para simulacao do EMSC2 - 2011

clc

global temp estado

//Incremento de tempo na resposta [s]

Dt = 0.0005;

// massas do problema [kg]

M=1.0;

m=9.0;

// comprimento da barra [m]

L = 1.5;

// metade do comprimento natural da mola [m]

r = 0.050;

// metade do comprimento do bloco [m]

a=0.025;

// fator beta

betah=0.25;

//rigidez da mola [N/m]

k=100.0;

// coeficiente de restituicao

e = 0.5;

// aceleracao da gravidade [m/s^2]

grav = 10.0;

// parametros derivados Jc e d

// Jc eh momento de inercia da barra com polo em C

// d eh a distancia de C a G na barra

Jc= m*L*L*(1/3+betah*betah-betah);

d= L*(0.5 -betah);

tol=1.0e-10;

// carrega definicao da funcao f

// que contem as equacoes de movimento

exec eq_mov.sci;

// estado Inicial

y0=[2.0*r+a;0.0;%pi/3;0.0];

// instante final de integracao [s]

t_final = 12.0;

// aloca memoria para vetor tempo

n_tempo=floor(t_final/Dt);

for i=1:n_tempo

temp(i)=(i-1)*Dt;

end

// inicio da integracao

estado(:,1)=y0;

for i=1:n_tempo-1

time(1)=temp(i);

time(2)=temp(i+1);

[est,w,q]=ode("adams",y0,time(1),time,f);

y0=est(:,2);

if ((est(1,2)<tol)&(est(2,2)<-tol)) then

y0(2)=-e*est(2,2);

y0(4)=est(4,2)-m*d*est(2,2)*cos(est(3,2))*(1+e)/Jc;

end

if ((4*r-2*a-est(1,2)<tol)&(est(2,2)>tol)) then

y0(2)=-e*est(2,2);

y0(4)=est(4,2)-m*d*est(2,2)*cos(est(3,2))*(1+e)/Jc;

end

if ((est(1,2)<tol)&(est(2,2)==0.0)) then

y0(2)=est(2,2);

y0(4)=est(4,2);

end

if ((4*r-2*a-est(1,2)<tol)&(est(2,2)==0.0)) then

y0(2)=est(2,2);

y0(4)=est(4,2);

end

estado(:,i+1)=est(:,2);

end

// elabora um grafico do deslocamento x(t)

// poderia ser dx/dt(t) ou teta(t) ou d(teta)/dt(t)

scf(0);

xtitle(’Posicao X(t) (m) vs. Tempo (s)’);

plot2d(temp,estado(1,:));

Na primeira parte do script os parâmetros do problema são informadas,
como variáveis globais. Define-se o tempo inicial, o tempo final de integração,
o intervalo de tempo em que gostaŕıamos de conhecer o estado do sistema
Dt, e um vetor de tempos a altura inicial da part́ıcula h0, a velocidade inicial
da part́ıcula v0, o estado inicial y0, o coeficiente de restituição e a aceleração
da gravidade grav.

Como o integrador deve ser interrompido na condição de colisão, a cada
intervalo Dt são verificadas as condições de choque. É necessário ir armaze-
nando os estados já calculados em uma matriz perene, que conterá a história
dos estados de diversos peŕıodos sem colisão. Definem-se então dois vetores,
temp e a matriz estado, onde cada coluna irá conter o estado num certo
instante de tempo (i− 1) ∗Dt. Um gráfico da história do deslocamento x(t)

está na fig. 5.2.

Figura 5.2: Barra articulada em anel

5.3 O critério de colisão

Se a distância a um batente for muito pequena e a velocidade conduz à
colisão, então, a condição de colisão foi satisfeita e a integração deve ser
re-inicializada com novas condições de velocidade, resultado do choque.

/* condicao de choque em P */

if ((est(1,2)<tol)&(est(2,2)<-tol)) then

y0(2)=-e*est(2,2);

y0(4)=est(4,2)-m*d*est(2,2)*cos(est(3,2))*(1+e)/Jc;

end

/* condicao de choque em Q */

if ((4*r-2*a-est(1,2)<tol)&(est(2,2)>tol)) then

y0(2)=-e*est(2,2);

y0(4)=est(4,2)-m*d*est(2,2)*cos(est(3,2))*(1+e)/Jc;

end

/* distancia pequena mas velocidade nula */

if ((est(1,2)<tol)&(est(2,2)==0.0)) then

y0(2)=est(2,2);

y0(4)=est(4,2);

end

/* distancia ao batente Q pequena, mas velocidade nula */

if ((4*r-2*a-est(1,2)<tol)&(est(2,2)==0.0)) then

y0(2)=est(2,2);

y0(4)=est(4,2);

end

5.4 Erro no conhecimento do instante de cho-

que

A estratégia de integração proposta neste script é aproximada. Perde-se o
instante exato em que o choque ocorre. Se o valor do incremento temporal
Dt para verificação do choque for diminuido, o erro a respeito do instante
de choque diminui. No caṕıtulo 4 há um procedimento para determinar com
precisão o instante do choque.

Sempre que inicializado o integrador ode() responde com uma lista de
informações, uma matriz com a história de estados Y nos instantes de tempo
definidos previamente pelo vetor temp. A estratégia de integração de movi-
mento com eventuais choques consiste em integrar as equações de movimento
em pequenos intervalos de cada vez. Ao final de cada intervalo, verificam-se
as condições de choque. Se ocorreu choque, o campo de velocidades deve ser
modificado. A seguir retoma-se a integração em um curto intervalo de tempo
Dt.


